

The Sorcerer's Apprentice cont.  
by Wayne A. Schotten

#### TEXT SCREEN CONTROL IN BASIC

If you depend only on the Apple manuals for Business BASIC, you may not be aware of the fact that you have three different text screen modes available. BizBASIC starts up with an 80 column B&W text screen, but you can also use a 40 column B&W or a 40 column color text screen. This appears to be a by-product of Apple II emulation mode.

What degenerate would use an outmoded, home computer standard like 40 columns? Me! We programming perverts want menu screens that are easy to read, perhaps even colorful.

To change text modes, you send characters to the console with PRINT CHR\$(nnn) function. ASCII character code (16) sets the text mode, (19) sets foreground color, and (20) sets the background color. By sending CHR\$(16) followed by a mode number, will set up your screen the way you like.

Mode 0: 40 column B&W  
Mode 1: 40 column color  
Mode 2: 80 column B&W

(The BASIC word "TEXT" will also reset your screen to 80 column B&W.)

Likewise, text color and background color are sent by sending the control character number followed by an integer from 0 to 15. (Check your manual for a color list.) For example, a line in BizBASIC to set 40 column color would be

```
960 PRINT CHR$(16);CHR$(1)
```

CHR\$(28) will clear the current screen, the same as the BASIC word "HOME". Other useful control characters (but by no means a complete list) are

(07) beep  
(14) turn the screen off like  
Control-5  
(15) turn the screen on  
(16) same as BASIC word "NORMAL"  
(18) same as BASIC word "INVERSE"

If you are issuing a series of screen controls, you may prefer to include (16) or (17) rather than use a separate line for "NORMAL" or "INVERSE".

A complete menu set-up might look like this:

```
10 PRINT CHR$(16);CHR$(1);CHR$(19);CHR$(13);CHR$(20);CHR$(4)
20 HOME
30 PRINT DATE$; TAB(18);"MENU"; TAB(32); TIME$
40 HPOS=10:VPOS=4
50 PRINT"0 First Selection on Menu"
60 HPOS=10:VPOS=6
70 PRINT"1 Last Selection on Menu"
80 VPOS=24
90 PRINT"Select by Number or Press ";CHR$(127);"Q to Reboot"
100 GET q$
110 IF ASC(q$)=241 THEN BYE
120 IF q$="0" THEN GOTO 110
130 IF q$<>"1" THEN GOTO 80:ELSE GOTO 110
140 TEXT:HOME:END
```

(If you're wondering just what in blue blazes is CHR\$(127) or similiarly (255), in line 90, sorry, sorry, sorry I'm not going to tell you. You'll just have to try it for yourself.)

Some color combinations are poor or even unreadable. This program will show you all combinations of colored text on colored backgrounds.

```
10 PRINT CHR$(16);CHR$(1)
20 FOR foreground%=0 TO 15
30   FOR background%=0 TO 15
40     PRINT CHR$(20);CHR$(background%)
50     PRINT CHR$(19);CHR$(foreground%)
60     FOR i%=0 TO 31
70       PRINT
80       NEXT
90     HPOS=5:VPOS=8
100    PRINT"Foreground is ";foreground%
110    HPOS=5:VPOS=10
120    PRINT"Background is ";background%
130    HPOS=1:VPOS=12
140    PRINT"THE LAZY DOGS JUMPED OVER THE QUICK"
150    PRINT"BROWN FOX"
160    PRINT"The lazy dogs jumped over the quick"
170    PRINT"brown fox"
180    FOR i%=1 TO 1000:NEXT
190    NEXT background%
200  NEXT foreground%
210 TEXT:HOME:LIST
```

There is much more we could discuss concerning screen control, but that'll be another article. Next issue, I'll be discussing gameports A and B. I want to hear from you, whether new to programming or an old hand. Wayne Schotten, %Precison Audio, Pier 26, San Francisco, CA 94105, 415-541-0960.